Programming Note

Agilent Technologies Introductory Programming Guide For the 8757D/E Scalar Network Analyzer with the HP 9000 Series 200/300 Desktop Computer (BASIC)

Agilent Technologies

Manufacturing Part Number: 08757-90116

Printed in USA

Print Date: July 1992

08757-90116

Hewlett-Packard to Agilent Technologies Transition

This manual may contain references to HP or Hewlett-Packard. Please note that Hewlett-Packard's former test and measurement, semiconductor products and chemical analysis businesses are now part of Agilent Technologies. To reduce potential confusion, the only change to product numbers and names has been in the company name prefix: where a product number/name was HP XXXX the current name/number is now Agilent XXXX. For example, model number HP8648 is now model number Agilent 8648.

Documentation Warranty

DFARS/Restricted Rights Notice

Printing Copies of Documentation from the Web

To print copies of documentation from the Web, download the PDF file from the Agilent web site:

- Go to http://www.agilent.com.
- Enter the document's part number (located on the title page) in the **Quick Search** box.
- Click GO.
- Click on the hyperlink for the document.
- Click the printer icon located in the tool bar.

Contacting Agilent

This information supersedes all prior HP contact information.			
Online assistance: www.agilent.com/find/assist			
Americas			
Brazil *(tel)* (+55) 11 3351 7012 *(fax)* (+55) 11 3351 7024	**Canada** *(tel)* 888 447 7378 *(fax) 905 282 6495*	**Mexico** *(tel)* 1 800 254 2440 *(fax)* 1 800 254 4222	**United States** *(tel)* 800 829 4444 *(alt)* (+1) 303 662 3998 *(fax)* 800 829 4433
Asia Pacific and Japan			
Australia (*tel*) 1 800 225 574 (*fax*) 1 800 681 776 (*fax*) 1 800 225 539	**China** (*tel*) 800 810 0508 (*alt*) 800 810 0510 (*fax*) 800 810 0507 (*fax*) 800 810 0362	**Hong Kong** (*tel*) 800 933 229 (*fax*) 800 900 701	**India** (*tel*) 1600 112 626 (*fax*) 1600 112 727 (*fax*) 1600 113 040
Japan (Bench) (*tel*) 0120 32 0119 (*alt*) (+81) 426 56 7799 (*fax*) 0120 01 2144	**Japan (On-Site)** (*tel*) 0120 802 363 (*alt*) (+81) 426 56 7498 (*fax*) (+81) 426 60 8953	**Singapore** (*tel*) 1 800 275 0880 (*fax*) (+65) 6755 1235 (*fax*) (+65) 6755 1214	**South Korea** (*tel*) 080 778 0011 (*fax*) 080 778 0013
Taiwan (*tel*) 0800 047 669 (*fax*) 0800 047 667 (*fax*) 886 3492 0779	**Thailand** (*tel*) 1 800 2758 5822 (*alt*) (+66) 2267 5913 (*fax*) 1 800 656 336	**Malaysia** (*tel*) 1800 880 399 (*fax*) 1800 801 054	
Europe			
Austria (*tel*) 0820 87 44 11* (*fax*) 0820 87 44 22	**Belgium** (*tel*) (+32) (0)2 404 9340 (*alt*) (+32) (0)2 404 9000 (*fax*) (+32) (0)2 404 9395	**Denmark** (*tel*) (+45) 7013 1515 (*alt*) (+45) 7013 7313 (*fax*) (+45) 7013 1555	**Finland** (*tel*) (+358) 10 855 2100 (*fax*) (+358) (0) 10 855 2923
France (*tel*) 0825 010 700* (*alt*) (+33) (0)1 6453 5623 (*fax*) 0825 010 701*	**Germany** (*tel*) 01805 24 6333* (*alt*) 01805 24 6330* (*fax*) 01805 24 6336*	**Ireland** (*tel*) (+353) (0)1 890 924 204 (*alt*) (+353) (0)1 890 924 206 (*fax*)(+353) (0)1 890 924 024	**Israel** (*tel*) (+972) 3 9288 500 (*fax*) (+972) 3 9288 501
Italy (*tel*) (+39) (0)2 9260 8484 (*fax*) (+39) (0)2 9544 1175	**Luxemburg** (*tel*) (+32) (0)2 404 9340 (*alt*) (+32) (0)2 404 9000 (*fax*) (+32) (0)2 404 9395	**Netherlands** (*tel*) (+31) (0)20 547 2111 (*alt*) (+31) (0)20 547 2000 (*fax*) (+31) (0)20 547 2190	**Russia** (*tel*) (+7) 095 797 3963 (*alt*) (+7) 095 797 3900 (*fax*) (+7) 095 797 3901
Spain (*tel*) (+34) 91 631 3300 (*alt*) (+34) 91 631 3000 (*fax*) (+34) 91 631 3301	**Sweden** (*tel*) 0200 88 22 55* (*alt*) (+46) (0)8 5064 8686 (*fax*) 020 120 2266*	**Switzerland (French)** (*tel*) 0800 80 5353 opt. 2* (*alt*) (+33) (0)1 6453 5623 (*fax*) (+41) (0)22 567 5313	**Switzerland (German)** (*tel*) 0800 80 5353 opt. 1* (*alt*) (+49) (0)7031 464 6333 (*fax*) (+41) (0)1 272 7373
Switzerland (Italian) (*tel*) 0800 80 5353 opt. 3* (*alt*) (+39) (0)2 9260 8484 (*fax*) (+41) (0)22 567 5314	**United Kingdom** (*tel*) (+44) (0)7004 666666 (*alt*) (+44) (0)7004 123123 (*fax*) (+44) (0)7004 444555		
(tel) = primary telephone number; *(alt)* = alternate telephone number; *(fax)* = FAX number; * = in country number 8/03/04			

HP–IB Programming Note

HEWLETT PACKARD

Introductory Programming Guide

For the HP 8757D/E scalar network analyzer with the HP 9000 series 200/300 desktop computer (BASIC)



Introduction

This programming note describes the remote operation of the HP 8757D/E Scalar Network Analyzer with the HP 9000 Series 200/300 desktop computer used as a controller. This includes the HP Vectra PC configured with the HP 82300D measurement coprocessor or HP 832324B high performance measurement coprocessor. Included in this guide are several short programs that demonstrate the use of the analyzer with HP–IB commands, and a diagram of system connections for remote control.

The HP 8757D/E is a fully programmable analyzer capable of making magnitude–only transmission and reflection measurements over an RF and microwave frequency range of 10 MHz to 110 GHz. When used with an HP–IB computer, the analyzer's front panel may be remotely controlled, along with most softkey functions and some functions accessible only via HP–IB. The analyzer exerts control over a source (HP 8350B, HP 8340B/41B or HP 8360 series), digital plotter (HP 7440A or HP 7550A), and printer (HP 2225A ThinkJet, HP 3630A PaintJet, or HP 2227B QuietJet Plus) connected to the 8757 SYSTEM INTERFACE.

This note assumes you are familiar with front panel (local) operation of the HP 8757D/E. If not, refer to the operating manual. You should also be familiar with the HP 9000 Series 200/300 computer, particularly HP–IB operation. Throughout the rest of this document, the term *computer* refers to any of these computers.

Reference information

The following texts provide additional information on the HP Interface Bus, the analyzer, the source, or the computer. See "Replaceable Parts" for ordering information.

HP 8757 literature

- HP 8757D Operating Manual.
- HP 8757C/E Operating Manual.
- Programming Note: *Quick Reference Guide for the HP 8757C/E Scalar Network Analyzer.*

Source literature

- Programming Note: *Introductory Operating Guide for the HP 8350B Sweep Oscillator with the HP 9000 Series 200 Computers (BASIC).*
- Programming Note: *Quick Reference Guide for the HP 8350B Sweep Oscillator.*
- Programming Note: *Introductory Operating Guide for the HP 8340A Synthesized Sweeper with the HP 9000 Series 200 Computers (BASIC).*
- Programming Note: *Quick Reference Guide for the HP 8340B Synthesized Sweeper.*
- Product Note: *HP 8340/41 to HP 8360 System Conversion Guide.*
- *HP 8360 Series Synthesized Sweepers Quick Reference Guide.*
- *HP 8360 Operating and Programming Reference Manual.*

HP 9000 series 200/300 computer literature

- BASIC Operating Manual.
- BASIC Programming Techniques.
- BASIC Language Reference.
- BASIC Interfacing Techniques.
- BASIC Graphics Techniques.

General HP−IB literature

- Condensed Description of the Hewlett−Packard Interface Bus.
- Tutorial Description of the Hewlett−Packard Interface Bus.

Equipment required

1 HP 8757D/E Scalar Network Analyzer.

1 HP 8350B Sweep Oscillator with plug−in or HP 8340B/41B Synthesized Sweeper or HP 8360 Series Synthesized Sweeper.

1 HP 9000 Series 200/300 Computer with BASIC 5.0 or higher, and at least 64K bytes of free user memory.

1 HP 85027A/B/C/D/E Directional Bridge.

1 HP 11664A/E Detector or HP 85025A/B/D/E Detector with connector type to match bridge and test device.

1 Shielded open circuit with connector to mate with bridge.

1 Short circuit with connector to mate with bridge.

3 HP 11170C BNC cables, 122 cm (48 in.). (4 are needed with HP 8340B/41B).

2 HP 10833A/B/C/D HP−IB cables.

1 Test device.



Figure 1. System Connections

Set–up

Connect the instruments as shown in figure 1. The following procedure sets the HP–IB addresses of the instruments to operate properly with the programs contained in this guide.

1. Turn on the HP 8350B Sweep Oscillator. Press **[SHIFT]** **[LCL]**. The FREQUENCY/TIME display shows the current HP–IB address of the source. If it is not 19, press **[1]** **[9]** to set the address to 19. The HP 8340B or 8341B Synthesized Sweeper operates the same, although the address is displayed in the right–hand display area.
2. Power on the HP 8757D/E Scalar Network Analyzer. The current HP–IB address is shown in the active entry area of the CRT. If it is not 16, press **[LOCAL]** *[8757]* **[1]** **[6]** **[ENT]** to set the address to 16.
3. Turn on your computer and your load BASIC operating system. Ensure that the following binary (BIN) programs are loaded: DISC, CS80, HPIB, GRAPH, ERR, and IO. LIST BIN lets you view the currently loaded binaries.

Check out procedure

1. Press **[PRESET]** on the analyzer. If the 8757 SYSTEM INTERFACE is properly connected, and the address of the source correctly set, both the analyzer and the source will perform an instrument preset. If either instrument detects a failure during instrument preset, that instrument displays the error encountered. The operating manual of the source gives instructions to help you interpret the error message. If the analyzer displays an error message, see "In Case of Difficulty."
2. To verify the HP–IB connections made between the analyzer and the computer, perform the following:

   Type "REMOTE 716" and press **[EXECUTE]**.

The R (remote) and L (listen) lights in the analyzer INSTRUMENT STATE area will light. The analyzer has received its HP–IB listen address.

Programming examples

In the following sections, example programs introduce the HP–IB capabilities of the analyzer. Each example program consists of these sections:

1. A description of the functions exercised.
2. The program listing.
3. An explanation of each program line.
4. Detailed instructions for operating the program.

When you finish all of the example programs, you will have a good idea of the power of the HP 8757D/E when used in an automatic system.

Program 1: remote, local, and local lockout

The analyzer may be used with the front panel (local operation) or programmed via HP–IB (remote operation). The programmer of the instrument system has control over the operation of all instruments in the system.

When the computer first addresses an instrument, the instrument is placed in a special remote operating mode, called remote mode. When in remote, the instrument does not respond to its front panel, except for the **[LOCAL]** key. **[LOCAL]**, when pressed, cancels the remote mode and allows the instrument to be used with its front panel.

The computer can also return the instrument to local operation. To do so, the computer sends a special command that forces the instrument to go to local mode.

Occasionally, the programmer of an automatic system needs to prevent the instrument operator from returning the instrument to local operation (via **[LOCAL]**). When the local lockout function of the computer is used, the instruments are prevented from exiting remote mode, even when **[LOCAL]** is pressed.

Frequently, the programmer needs to place the instruments connected to the computer into a known state. When preset, the analyzer defaults to the conditions shown below. The instrument preset function operates the same as the front panel **[PRESET]** key on the analyzer and the source. When presetting the analyzer and source, send the PRESET command only to the analyzer. The analyzer will preset the source attached to the 8757 SYSTEM INTERFACE.

HP 8757C/E instrument preset conditions

Channels 1 and 2 on. The channel menu appears in the softkey label area of the CRT.

- Measure power A on channel 1.
- Measure power B on channel 2.
- Measure power C[2] (or B) on channel 3[1].
- Measure power R on channel 4[1].
- Display measurement data in log magnitude format.
- Scale =20 dB/div.
- Reference level 0 dB for all channels.
- Reference level step size =20 dB.
- Averaging off.
- Averaging factor =8.
- Cursor off.
- All labels on.

1. HP 8757D only.
2. HP 8757D Option 001 only.

- Channel 1 as the active channel.
- Modulation drive on.
- Number of points =401.
- Detector mode set for AC detection.
- Smoothing set for 5.0% of span (off).
- Cursor format = log magnitude.
- Search value = −3 dB[1].
- Adaptive normalization off[1].
- Temperature compensation on.
- Repeat autozero off.
- Detector offset reset to zero.[2]
- Detector frequency off, start and stop set to 50 MHz.[1]

Source

- Instrument preset.
- Sweep time set to 200 ms.
- HP 8350B square wave modulation on.
- HP 8340/41 SHIFT PULSE on; RF Output on.
- HP 8360 scalar modulation on; RF output on; ramp sweep mode; analyzer mode.

Plotter

- Abort plot if in progress.
- P1 and P2 scaling points unchanged.
- Selection of plotter pens unchanged.

Printer

- Abort plot if in progress.

Disk drive[1]

- Aborts any data transfers in progress.
- Unit number unchanged.
- Volume number unchanged.
- ASCII or binary mode unchanged.

The following analyzer conditions are not changed during a PRESET (IP) command execution:

- Reference position.
- Trace memory.
- Save/Recall registers.
- HP−IB addresses.
- Request mask.
- Limit lines[1].
- Title.
- User−defined plot.
- 8757 System Interface control on/off.
- Repeat autozero timer.
- Display intensity.
- Display colors[1].

1. HP 8757D only.
2. Detector offset remains unchanged for HP 8757E.

Program 1 listing

```
10   ASSIGN @Sna TO 716
20   ABORT 7
30   CLEAR @Sna
40   REMOTE @Sna
50   PAUSE
60   REMOTE @Sna
70   LOCAL LOCKOUT 7
80   PAUSE
90   LOCAL 7
100  PAUSE
110  OUTPUT @Sna;"IP"
120  END
```

Program 1 explanation

Line 10	Assign the address of the HP 8757D/E to an I/O path. This is not required, but it is good programming practice. If you change the address of the instrument later, you only change the address in one place in your program.
Line 20	Abort any HP−IB transfers and reset the computer's HP−IB interface.
Line 30	Clear the analyzer's HP−IB interface.
Line 40	Set the analyzer and source to remote mode.
Line 50	Temporarily stop execution.
Line 60	Set the analyzer and source to remote mode.
Line 70	Lock out the **[LOCAL]** key of the analyzer and source.
Line 80	Temporarily stop execution.
Line 90	Set the analyzer and source to local mode.
Line 100	Temporarily stop execution.
Line 110	Preset the analyzer and source.
Line 120	End program execution.

Running program 1

1. Press **[SHIFT]** **[RESET]** on the computer. Type "SCRATCH" and press **[EXECUTE]**. This clears the program memory of the computer.
2. Type in the program.
3. Press **[RUN]** on the computer.
4. When the program stops, the analyzer is in remote mode. You can verify this by observing the lights in the INSTRUMENT STATE area of the analyzer. The R (remote) and L (listen) lights should be on. Try pressing any key on the analyzer (except [LOCAL]). Nothing happens. The source is also in remote mode. Now press **[LOCAL]** and verify that the keys on the analyzer are active. Also, notice the R light went out when you pressed **[LOCAL]**. The source went into local mode along with the analyzer.

5. Press **[Continue]** on the computer. The analyzer is again in remote mode. This time, however, the [LOCAL] key is locked out. Try pressing **[LOCAL]** and the other keys. None of the keys on the analyzer or the source cause any action.

6. Press **[Continue]** on the computer. All instruments on the HP−IB interface are returned to local mode, including the analyzer and source. To set only the analyzer into local mode, the LOCAL 716 command can be given from the computer. Verify that the R light on the analyzer is off and the REM light on the source is off.

7. Press **[Continue]** on the computer. The analyzer and source are both preset. Note that the computer sent the instrument PRESET command only to the analyzer. The analyzer, in turn, preset the source.

Remember, to preset both the analyzer and the source, you only need to send the instrument PRESET command to the analyzer. Do not send instrument PRESET to the source by way of passthru mode (discussed in program 3).

Program 2: controlling the front panel

All front panel keys and most of the softkeys of the analyzer may be programmed remotely via HP−IB. For example, you can program the scale per division, reference level, and reference position for each channel.

Program 2 listing

```
10   ASSIGN @Sna TO 716
20   ABORT 7
30   CLEAR @Sna
40   OUTPUT @Sna;"IP"
50   PAUSE
60   OUTPUT @Sna;"C1C0C2"
70   PAUSE
80   OUTPUT @Sna;"SD10"
90   PAUSE
100  OUTPUT @Sna;"RL−10"
110  PAUSE
120  OUTPUT @Sna;"RP4"
130  PAUSE
140  OUTPUT @Sna;"IA"
150  PAUSE
160  OUTPUT @Sna;"C0C1 SD5; RP4; RL−5"
170  END
```

Program 2 explanation

Line 10 Assign an I/O path to the HP−IB address of the HP 8757D/E.

Line 20 Abort any transfers and clear the HP−IB interface of the computer.

Line 30 Clear the HP−IB interface of the analyzer.

Line 40 Preset the analyzer and the source.

Line 50 Temporarily stop execution.

Line 60 Select channel 1 and turn it off. Turn channel 2 on.

Line 70 Temporarily stop execution.

Line 80 Set the scale per division to 10 dB. No terminator (;) is needed because this is the last command in the statement.

Line 90 Temporarily stop execution.

Line 100 Set the reference level to −10 dBm. Again, note the absence of a terminator (;).

Line 110 Temporarily stop execution.

Line 120 Set the reference position line to the center of the CRT (4th graticule). No terminator is needed because this is the last command on the line.

Line 130 Temporarily stop execution.

Line 140 Program channel 2 to measure reflection (input A) instead of transmission (input B).

Line 150 Temporarily stop execution.

Line 160 Many commands on one line, with terminators. Turn channel 2 off (C2C0) and channel 1 on (C1). Set the scale per division (SD) to 5 dB, the reference position line (RP) to the center of the CRT, and the reference level (RL) to −5 dBm.

Line 170 End execution.

Running program 2

1. Type "SCRATCH" and press **[EXECUTE]** on the computer. This erases the previous program.
2. Type in this program and press **[RUN]** on the computer.
3. The computer presets the analyzer and source and pauses. Note the settings of channel 1 and 2, then press **[Continue]**.
4. Channel 1 is turned off. Channel 2 is now the active channel, as you can see from the highlighted box around the channel 2 mode labels on the analyzer CRT. Press **[Continue]**.
5. Channel 2 scale per division is now set to 10 dB. It defaulted to 20 dB/div at preset. Press **[Continue]**.
6. The reference level is set to −10 dBm (it was 0.0 dBm). Press **[Continue]**.
7. The reference position line is set to the center of the CRT (graticule 4). The top of the CRT is graticule 8 and the bottom is graticule 0. Press **[Continue]**.
8. Change the measurement to reflection (input A), instead of transmission (input B). At preset, channel 2 defaults to input B. Press **[Continue]**.
9. In one statement: turn off channel 2, turn on channel 1, set the scale per division to 5 dB, set the reference position line to the center of the CRT, and set the reference level to −5 dBm.

NOTE: The semicolon (;) terminators are needed after any analyzer command that can have a variable length. Extra terminators never hurt, so use them liberally.

Program 3: passthru mode

In normal operation, the system source, digital plotter, printer, and disk drive (HP 8757D only) are connected to the 8757 SYSTEM INTERFACE. This connection allows the analyzer to control and extract information from the other parts of the measurement system. To allow you to control the source and plotter with the computer, the analyzer has a built–in PASSTHRU command that takes a command from the computer and passes it on to one of the instruments connected to the 8757 SYSTEM INTERFACE.

To initiate passthru mode, first tell the analyzer which instrument you wish to command by setting the passthru address. Then, to talk (or listen) to that device, address the analyzer's special passthru HP–IB address (which is different from the analyzer's HP–IB address). While in the passthru mode, the analyzer stops updating its CRT and does not respond to its front panel (because it's in remote mode). To remove the analyzer from passthru mode, simply address it via HP–IB. While in passthru mode, do not press **[LOCAL]** on the analyzer.

The analyzer's passthru address is calculated from its HP–IB address. If the address of the analyzer is even (such as 16 decimal) then the passthru address is the next larger number (17 decimal). If the address of the analyzer is odd (such as 15 decimal), then the passthru address is the next smaller number (14 decimal). Never set the address of the analyzer such that its address conflicts with one of the instruments connected to the 8757 SYSTEM INTERFACE. For instance, if the source is set to 19 decimal, do not set the address of the analyzer to 19.

Data can be sent to or received from any instrument on the 8757 SYSTEM INTERFACE via passthru mode. LOCAL, REMOTE, and TRIGGER HP–IB messages do not pass through the analyzer.

Program 3 listing

```
10   PRINTER IS 1
20   ASSIGN @Sna TO 716
30   ASSIGN @Passthru TO 717
40   ABORT 7
50   CLEAR @Sna
60   OUTPUT @Sna;"IP"
70   OUTPUT @Sna;"PT19"
80   OUTPUT @Passthru;"OPFA"
90   ENTER @Passthru;Min_freq
100  Min freq=Min freq/1.E+9
110  OUTPUT @Passthru;"OPFB"
120  ENTER @Passthru;Max_freq
130  Max_freq=Max_freq/1.E+9
140  OUTPUT @Sna
150  PRINT "Frequency limits:";Min freq;"to";Max freq;"GHz"
160  INPUT "Start_frequency (GHz)?",Start_freq
170  INPUT "Stop_frequency (GHz)?",Stop_freq
180  OUTPUT @Passthru;"FA";Start_freq;"GZ
     FB";Stop_freq;"GZ"
190  OUTPUT @Sna
200  END
```

Program 3 explanation

Line 10	Direct the printed output to the computer CRT.
Line 20	Assign an I/O path to the address of the analyzer. (This is the analyzer's control address).
Line 30	Assign an I/O path to the analyzer's passthru address. By communicating to this HP–IB address, the computer will control a device connected to the 8757 SYSTEM INTERFACE.
Line 40	Abort any transfers and clear the HP–IB interface of the computer.
Line 50	Clear the HP–IB interface of the analyzer.
Line 60	Preset the analyzer and source.
Line 70	Tell the analyzer which device is controlled through the analyzer's passthru address. In this case, the source (device 19).
Line 80	Send a command to the source. Command it to output its current start frequency.
Line 90	Read the start frequency from the source.
Line 100	Scale the start frequency to display it in GHz.
Line 110	Command the source to output its current stop frequency.
Line 120	Read the stop frequency from the source.
Line 130	Scale the stop frequency to display it in GHz.
Line 140	Exit passthru mode by addressing the analyzer.
Line 150	Print the start and stop frequency.
Line 160	Get the start frequency from the user.
Line 170	Get the stop frequency from the user.
Line 180	Set the start and stop frequency of the source to those given by the user.
Line 190	Exit passthru mode by addressing the analyzer.
Line 200	End program execution.

Running program 3

1. Clear the program memory of the computer and type in the program.
2. Press **[RUN]** on the computer.

3. The computer presets the analyzer and the source, reads the start and stop frequency of the source, and displays it on the CRT of the computer. At preset, the source defaults to the full frequency range of the plug–in. The values read, then, represent the frequency limits of this plug–in. When the computer stops, it displays the prompt:

   `Start frequency (GHz)?`

   Enter a start frequency in the frequency range of the plug–in and press **[Continue]**.

4. The computer displays the prompt:

   `Stop frequency (GHz)?`

   Enter a stop frequency in the frequency range of the plug–in (but higher than the start frequency) and press **[Continue]**.

5. The computer sets the start and stop frequency of the source to those you entered. The analyzer immediately begins sweeping the frequency range you defined.

6. Try deleting or "commenting out" line 190 in the program. Now, when the program ends, the analyzer shows the message `DATA PASSTHROUGH EXECUTING` and the display is frozen (not sweeping). To exit passthru mode, type "OUTPUT 716" and press **[EXECUTE]** on the computer. The analyzer displays `DATA PASSTHROUGH COMPLETE` and begins sweeping.

Points to remember: You must address the analyzer after using passthru mode to return it to normal swept operation. Any command can be sent via passthru mode to any instrument on the 8757 SYSTEM INTERFACE, and any data can be read. Service requests and parallel polls do not pass through the analyzer.

Program 4: cursor operations

To enhance the speed and accuracy of measurements, the analyzer contains a built–in cursor that displays the frequency and magnitude of a trace at any given point. To make measurements even more efficient, the cursor may be set to the maximum or minimum point on the trace simply by pressing a softkey. These cursor functions are available via HP–IB commands.

With a computer, the cursor may be turned on and off, its position (0 to n–1, where n is the number of points per trace) set, its value and position read, and set to the maximum or minimum point on the trace. The cursor functions all apply to the active channel. You have complete control over cursor operations via HP–IB.

Cursor programming is especially useful for measuring parameters like flatness and maximum power, where you are interested in the highest and lowest point on the trace. For measuring parameters such as 3 dB points and other specific points (not a maximum or minimum), it is more efficient to use either the cursor search functions (available on the HP 8757D only) or to read the entire trace and search for the points you need.

Program 4 listing

```
10   PRINTER IS 1
20   Start_freq=.01
30   Stop_freq=20
40   ASSIGN @Sna TO 716
50   ASSIGN @Passthru TO 717
60   ABORT 7
70   CLEAR @Sna
80   OUTPUT @Sna;"IP"
90   OUTPUT @Sna;"PT19"
100  OUTPUT @Passthru;"FA";Start_freq;"GZ
     FB";Stop_freq;"GZ"
105  OUTPUT@Sna
110  OUTPUT @Sna;"C2 CXOC"
120  ENTER @Sna;Value,Posn
130  PRINT "Cursor reads ";Value;"dB at position";Posn
140  INPUT "Desired cursor position (0..400)?",New_posn
150  OUTPUT @Sna;"SC";INT(New_posn+.5)
160  OUTPUT @Sna;"OC"
170  ENTER @Sna;Value,Posn
180  PRINT "Value at position";Posn;"is ";Value;"dB."
190  INPUT "Cursor frequency (GHz)?",Cur_freq
200  New_posn=400*((Cur freq–Start_freq)/(Stop_freq–
     Start_freq))
210  OUTPUT @Sna;"SC";INT(New_posn+.5)
220  OUTPUT @Sna;"OC"
230  ENTER @Sna;Value,Posn
240  Cur_freq=Start_freq+(Stop_freq–Start_freq)*(Posn/400)
250  PRINT "Cursor reads ";Value;"dB at";Cur_freq;"GHz."
260  END
```

Program 4 explanation

Line 10	Direct the printed output to the computer CRT.
Line 20	Define the start frequency of the desired sweep in GHz.
Line 30	Define the stop frequency of the desired sweep in GHz.
Line 40	Assign an I/O path to the address of the analyzer.
Line 50	Assign an I/O path to the passthru address of the analyzer.
Line 60	Abort any transfers and clear the HP–IB interface of the computer.
Line 70	Clear the HP–IB interface of the analyzer.

Line 80 Preset the analyzer and source. This sets the number of points per trace to 401.

Line 90 Tell the analyzer which instrument is controlled through the passthru address (19 is the source).

Line 100 Command the source to set a start frequency of 0.01 GHz and a stop frequency of 20 GHz.

Line 105 Turn passthru mode off. Allow analyzer display to update.

Line 110 Set the cursor to the maximum point on channel 2 and command the analyzer to output the cursor's value and position.

Line 120 Read the value and position of the cursor.

Line 130 On the computer CRT, print the value and position of the cursor.

Line 140 Get the new cursor position from the user. Input should be between 0 and 400.

Line 150 Set the cursor to the new cursor position chosen by the user. The INT function truncates instead of rounding, so add 0.5 to the cursor position before making it an integer.

Line 160 Command the analyzer to output the cursor's value and position.

Line 170 Read the value and position of the cursor at its new position.

Line 180 Print the cursor's value and position on the computer CRT.

Line 190 Get the new cursor frequency from the user. It must be within the frequency range of the sweep selected.

Line 200 Calculate the position of the cursor from its frequency and the start and stop frequencies of the current measurement.

Line 210 Set the cursor to the desired position.

Line 220 Command the analyzer to output the cursor's value and position.

Line 230 Read the cursor's value and position.

Line 240 Calculate the cursor's actual frequency from its position and the start and stop frequencies of the current measurement. You can easily program other start and stop frequencies by following the example in program 3.

Line 250 On the computer CRT, print the value and actual frequency of the cursor.

Line 260 End program execution.

Running program 4

1. Clear the program memory of the computer and type in the program.
2. Press **[RUN]** on the computer.
3. The computer turns on both channels and sets channel 1 to reflection (input A) and channel 2 to transmission (input B). The cursor is positioned to the maximum point on the channel 2 trace, and its value and position are read and displayed. At preset, the number of points per trace is 401.
4. The computer displays the prompt:

   `Desired cursor position (0..400)?`

   Type in a number between 0 and 400 and press **[Continue]**. A position of 0 represents the left side of the analyzer's CRT (lowest frequency) and 400 represents the right side of the CRT (highest frequency). The position is set, and the cursor's value and position are read and printed on the CRT of the computer.
5. The computer stops and displays the prompt:

   `Cursor frequency (GHz)?`

Enter a frequency within the current start and stop frequencies of the measurement (0.01 to 20 GHz). The nearest cursor position is calculated and set. The cursor's value and position are read, and the actual cursor frequency is calculated from the cursor's position.

Note: The original desired frequency and the actual cursor frequency are usually different. Because there are only 401 possible cursor positions, some frequencies cannot be set exactly.

To use more points per trace when using the HP 8757D, modify line 80 to be "IP SP801" for 801 points. Then modify "400" in lines 140, 200, and 240, to "800".

Program 5: read a single value

Measurements often require that a single value be read at a CW frequency, particularly when extremely good frequency accuracy and resolution are required.

The analyzer is able to read and send a single reading of any measurement channel, via HP−IB, to the computer. The OUTPUT VALUE (OV) command operates on the active channel and causes the analyzer to send one reading of measurement data. Even when the analyzer is in normalized mode (MEAS−MEM), the OV command sends the measured, not the normalized, data. This command, like the OUTPUT DATA command in program 6, can operate with either ASCII or fast binary formats.

Program 5 listing

```
10   PRINTER IS 1
20   ASSIGN @Sna TO 716
30   ASSIGN @Passthru TO 717
40   ABORT 7
50   CLEAR @Sna
60   OUTPUT @Sna;"IP"
```

```
70   OUTPUT @Sna;“PT19”
80   OUTPUT @Sna;“SW0”
90   Freq=2
100  Freq_step=.1
110  OUTPUT @Passthru;“CW”;Freq;“GZ SF”;Freq_step;“GZ”
120  OUTPUT @Sna;“C1IA”
130  FOR I=1 TO 21
140     OUTPUT @Sna;“OV”
150     ENTER @Sna;Value
160     PRINT I;“: ”,Value;“dB at”;Freq;“GHz”
170     OUTPUT @Passthru;“CW; UP”
180     Freq=Freq+Freq_step
190  NEXT I
200  OUTPUT @Passthru;“FA2GZ FB4GZ”
210  OUTPUT @Sna;“SW1”
220  END
```

Program 5 explanation

Line 10 Direct the printed output to the computer CRT.

Line 20 Assign an I/O path to the address of the analyzer.

Line 30 Assign an I/O path to the passthru address of the analyzer.

Line 40 Abort any transfers and clear the HP−IB interface of the computer.

Line 50 Clear the HP−IB interface of the analyzer.

Line 60 Preset the analyzer and source.

Line 70 Tell the analyzer which instrument is controlled through the passthru address (19 is the source).

Line 80 Put the analyzer in non−swept mode. This step is necessary when you read single values. After receiving this command, the analyzer stops updating its display.

Line 90 Define a start frequency for further measurements (in GHz).

Line 100 Define a frequency increment (in GHz).

Line 110 Put the source into CW mode at the start frequency, and set its frequency step size to that of the frequency increment.

Line 120 Command the analyzer to measure reflection (input A) on channel 1. This statement also causes the analyzer to exit passthru mode.

Line 130 Make 21 measurements, at equally spaced CW frequencies.

Line 140 Command the analyzer to send the current reading of channel 1 (the active channel) to the computer. The reading is taken immediately.

Line 150 Read the value. In this instance, no format has been defined so the default format of ASCII is in effect.

Line 160 Print the measurement number, the reading, and the frequency on the computer CRT.

Line 170 Command the source to increment the CW frequency by the step size set earlier (line 90). This is a very fast way of setting a series of equally spaced frequencies.

Line 180 Increment the variable that contains the current frequency. This variable is only used for printing the current frequency at each repetition of the loop.

Line 190 End of the loop.

Line 200 Command the source to sweep from 2 to 4 GHz. The source exits CW mode and returns to start/stop mode.

Line 210 Command the analyzer to return to swept mode. The analyzer again updates the trace information on the CRT. This command also exits passthru mode.

Line 220 End program execution.

Running program 5

1. Clear the program memory of the computer and type in the program.
2. Press **[RUN]** on the computer.
3. The source frequency is set immediately to 2 GHz and the computer begins reading reflection (input A) on the analyzer and printing the measurements. After 21 readings, the program ends.

Program 6: trace transfer

One feature that sets the HP 8757D/E apart is its ability to transfer an entire measurement trace to a computer at very high speed. A complete, high−resolution (0.01 dB), 401−point measurement can be sent to the computer in 35 milliseconds (binary format) or 800 milliseconds (ASCII format). Transfer time will be less for fewer points per trace, and greater for more points per trace.

The analyzer gives you complete flexibility when reading measurement traces via HP−IB. You can read from the active channel and you can read the stored memory trace, the current measurement trace, or the normalized trace (measurement−minus−memory). In addition, the memory trace can be written back to the analyzer, allowing you to save and restore calibration traces via HP−IB.

With trace transfer measurements, some frequency resolution is sacrificed for measurement speed. The number of points per trace can be programmed to control the resolution across the frequency range being swept. If you are measuring a device that changes very rapidly with frequency, it is possible to miss very narrowband responses that occur between measurement points if the resolution is low. For these cases, the measurement should be made at a higher resolution. The trace transfer method of measurement is much faster than CW point−by−point measurements.

Program 6 listing

```
10   ASSIGN @Sna TO 716
20   ASSIGN @Fast_sna TO 716;FORMAT OFF
30   ABORT 7
40   CLEAR @Sna
50   OUTPUT @Sna;"IP"
60   DIM Ascii_dat(0:400)
70   INTEGER Bin_dat(0:400)
80   OUTPUT @Sna;"C1IA C2IB"
90   WAIT 1
100  OUTPUT @Sna;"FD0 C1OD"
110  ENTER @Sna;Ascii_dat(*)
120  OUTPUT @Sna;"C1WM";Ascii_dat(*)
130  PAUSE
140  OUTPUT @Sna;"FD1 C2OD"
150  ENTER @Fast_sna;Bin_dat(*)
160  OUTPUT @Sna USING "#,K";"C2WM"
170  OUTPUT @Fast_sna;Bin_dat(*)
180  PAUSE
190  FOR I=0 TO 400
200    Bin_dat(I)=(I MOD 100)
210  NEXT I
220  OUTPUT @Sna;"C2C0 C1MY"
230  OUTPUT @Sna USING "#,K";"FD1 WM"
240  OUTPUT @Fast_sna;Bin_dat(*)
250  OUTPUT @Sna;"AS"
260  END
```

Program 6 explanation

Line 10 Assign an I/O path to the address of the analyzer.

Line 20 Assign another I/O path to the address of the analyzer, to be used for fast binary transfers.

Line 30 Abort any transfers and clear the HP−IB interface of the computer.

Line 40 Clear the HP−IB interface of the analyzer.

Line 50 Preset the analyzer and the source. This sets the number of points per trace to 401.

Line 60 Dimension an array to hold a trace in ASCII format. An array is 401 elements (0 to 400, inclusive).

Line 70 Dimension an array to hold a trace in binary-format. It is also 401 elements.

Line 80 Set channel 1 to reflection (input A) and channel 2 to transmission (input B).

Line 90 Wait for the source to sweep a few times, to insure the traces contain valid data. When you command the analyzer to output a trace, it responds immediately.

Line 100 Set the format to ASCII and command the analyzer to output the channel 1 measurement trace data.

Line 110 Read the measurement trace. Note the use of an asterisk (*) to designate the entire array.

Line 120 Write the measured trace back to the trace memory of channel 1. Reading the measurement trace and storing it into the memory trace is equivalent to executing the MEAS _MEM function (SM).

Line 130 Temporarily stop program execution.

Line 140 Set the format to binary and command the analyzer to output its channel 2 measurement trace.

Line 150 Enter the measurement trace through the I/O path that suspends formatting. This technique is useful for reading data from the analyzer at the highest possible speed.

Line 160 Command the analyzer to accept the trace into its channel 2 memory. Note the suppression of the normal carriage return/line feed sequence by the "#,K" format. If the cr/lf isn't suppressed, the analyzer assumes the first data point is null.

Line 170 Send the trace to the analyzer, again through the I/O path that suspends formatting.

Line 180 Temporarily stop program execution.

Line 190 Set up a loop for all 401 measurement points read from the analyzer.

Line 200 Calculate an arbitrary function and fill the binary data array. This function has no particular meaning, but represents some special calibration data (such as a short/open average).

Line 210 End of the loop.

Line 220 Turn off channel 2 and command channel 1 to display the trace memory data.

Line 230 Set the format to binary (redundant, but good practice) and command the analyzer to accept the following trace to channel 1 memory. Again, suppress the cr/lf sequence at the end of the line.

Line 240 Write the trace to the memory through the I/O path that suspends formatting.

Line 250 Command the analyzer to autoscale the current display, which is the memory trace just written.

Line 260 End program execution.

Running program 6

1. Clear the program memory of the computer and type in the program.
2. Press **[RUN]** on the computer.
3. Watching the analyzer CRT, you will see **DATA DUMP TO HP–IB** when it begins sending trace data to the computer, and **DATA DUMP TO TRACE MEMORY** when the computer sends the data back. The transfer takes about 800 milliseconds each way (ASCII transfer).
4. Watching the analyzer CRT, press **[Continue]** on the computer. The computer again reads and writes a trace of data, and the analyzer displays the same messages. The transfer is very fast, about 35 milliseconds each way (binary format).
5. Press **[Continue]** on the computer. The computer calculates an arbitrary function and sends it to a trace memory of the analyzer, where it is autoscaled and displayed. This function (a sawtooth pattern) has no significance. It represents a special calibration trace, such as a short/open average. With a computer, the analyzer measurement system may be calibrated over several different frequency ranges and changed from one to another very quickly, without re–calibration.

When writing memory traces in ASCII format, be sure to set the analyzer to ratio or single–input measurements before sending the trace. If you wish to transfer a higher resolution trace, modify line 50 to be "IP SP801" for 801 points. Then modify the "400" in lines 60, 70, and 190, to "800".

Program 7: using the TAKE SWEEP command

To make measurements as quickly and efficiently as possible, it is often necessary to synchronize the source with the analyzer. The TAKE SWEEP command gives the analyzer the ability to command the source to make a specified number of complete sweeps (1 to 255). This command is especially useful when using the trace transfer method of reading data from the analyzer.

To use the TAKE SWEEP command, place the analyzer in non–swept mode (SW0). Then give the TAKE SWEEP command with the number of sweeps desired (TSd). At the end of the specified number of sweeps, the analyzer informs the computer of the completion of this operation by setting a bit in its status byte.

The computer can detect this event in two ways:

- Monitor the status byte continuously until the bit is set (polling).
- Let the analyzer generate a service request (SRQ) and interrupt the computer.

Table 1 is a diagram of the status bytes of the analyzer. It shows all of the bits that can be used to either monitor or interrupt the computer. In this program, bit 4 (decimal value 16) is used to signal operation complete (all of the sweeps specified by the TAKE SWEEP command have been completed.)

When you follow the take sweep command with an output statement, such as OUTPUT DATA (OD), the data is sent immediately, not after the instructed number of sweeps. The two approaches mentioned overcome this by letting us send the data at the end of the specified number of sweeps, notimmediately. A third approach is to use the sweep hold mode (SW2) instead of the non–swept mode (SW0). In this mode the analyzer will prevent any HP–IB operations until the completion of the TAKE SWEEP command.

Program 7 listing

```
10    DIM Ascii_dat(0:400)
20    ASSIGN @Sna TO 716
30    ASSIGN @Passthru TO 717
40    ABORT 7
50    CLEAR @Sna
60    OUTPUT @Sna;"IP"
70    OUTPUT @Sna;"PT19"
80    OUTPUT @Passthru;"ST250MS"
90    OUTPUT @Sna;"C2C0 IB"
100   OUTPUT @Sna;"SW0 CS RM16"
110   OUTPUT @Sna;"TS10"
120   Stat=SPOLL(@Sna)
130   IF BIT(Stat,4)=0 THEN 120
140   OUTPUT @Sna;"C1OD"
150   ENTER @Sna;Ascii_dat(*)
160   OUTPUT @Sna;"SW1"
170   PAUSE
180   OUTPUT @Sna;"SW0 CS RM16"
190   ON INTR 7 GOTO Srq_recv
200   ENABLE INTR 7;2
210   OUTPUT @Sna;"TS10"
220   GOTO 220
230 Srq_recv: !
240   Stat=SPOLL(@Sna)
250   OUTPUT @Sna;"RM0"
260   OUTPUT @Sna;"C1OD"
270   ENTER @Sna;Ascii_dat(*)
280   OUTPUT @Sna;"SW1"
290   END
```

Program 7 explanation

Line 10 Dimension an array large enough to hold a trace of data (401 points).

Line 20 Assign an I/O path to the address of the analyzer.

Line 30 Assign an I/O path to the passthru address of the analyzer.

Line 40 Abort any transfers and clear the HP–IB interface of the computer.

Line 50 Clear the HP–IB interface of the analyzer.

Line 60 Preset the analyzer and source.

Line 70 Tell the analyzer which device is controlled through the passthru address. Address 19 belongs to the sweeper.

Line 80 Set the source to 250 milliseconds per sweep.

Line 90 Turn off channel 2 of the analyzer and select transmission (input B) for display on channel 1.

Line 100 Put the analyzer into non–swept mode. Clear the status register of the analyzer. Set the request mask to 16 (bit 4) so that the analyzer will set bit 4 ("operation complete") at the completion of the TAKE SWEEP command. Table 1 has a description of all bits in the status bytes.

Line 110 Command the analyzer to take 10 sweeps.

Line 120 Wait for the 10 sweeps to completed by reading the analyzer status byte.

Line 130 Test the status byte to see if bit 4 is set. If it is, then 10 sweeps have been completed. If bit 4 is not set, then continue to read and test the status byte until it is set.

Line 140 Command the analyzer to output the channel 1 trace data.

Line 150 Read the trace data.

Line 160 Return the analyzer to swept mode. The display now updates continuously.

Line 170 Temporarily stop program execution.

Line 180 Put the analyzer into non–swept mode. Clear the status register of the analyzer. Set the request mask to 16 (bit 4, OPERATION COMPLETE) so that the analyzer will send the computer a service request (SRQ) at the completion of the TAKE SWEEP command. This is the same as in line 100 except we will look for interrupts this time.

Line 190 Define the routine to be executed when the SRQ is received from the analyzer. The label "Srq_recv" is equivalent to line 230.

Line 200 Turn on interrupts in the computer. Specifically, allow an HP–IB service request to interrupt the computer. See the BASIC Language Reference of the computer for more detail about HP–IB programming.

Line 210 Command the analyzer to take 10 sweeps.

Line 220 Wait for the SRQ from the analyzer by putting the computer into a tight loop. If a PAUSE statement were used, the computer would not respond to interrupts.

Line 230 The computer begins execution here after receiving the SRQ from the analyzer.

Line 240 Read the status byte of the analyzer. This action clears the SRQ flag of the analyzer.

Line 250 Disable interrupt generation from the analyzer.

Line 260 Command the analyzer to output the channel 1 data trace.

Line 270 Read the channel 1 trace.

Line 280 Return the analyzer to swept mode. The analyzer display begins updating continuously.

Line 290 End of execution.

Running program 7

1. Clear the program memory of the computer and type in the program.
2. Press **[RUN]** on the computer.
3. The computer first presets the analyzer and source. It then sets the source to 250 milliseconds per sweep, and sets the analyzer to display transmission on channel 1.
4. The computer commands the analyzer to take 10 sweeps and polls the analyzer status byte to determine when they were completed. The computer reads a trace from the analyzer. Just before the trace is sent, you should see the display "freeze" as the TAKE SWEEP command is completed.
5. Press **[Continue]**, and the computer again tells the analyzer to take 10 sweeps. This time the computer receives an interrupt after the last sweep. The computer sits in a loop (line 220) and waits until the analyzer signals completion of the TAKE SWEEP command. In this segment of the program, you should not see the display "freeze" at all. Immediately after it receives the interrupt, the computer puts the analyzer back into swept mode. This method of sensing the end of a TAKE SWEEP command via an interrupt is more time–efficient than the polling method previously used because the computer can be doing something else during the 10 sweeps.

Table 1. HP 8757D/E Status Byte Descriptions

STATUS BYTE (#1)								
BIT #	7	6	5	4	3	2	1	0
Decimal Value	128	64	32	16	8	4	2	1
Function	N/A	Request Service (SRQ)	SRQ on HP–IB Syntax Error	SRQ on Operation Complete (Sweep, Plot or Print)	SRQ on Softkey Only Pressed	SRQ on Change in Extended Status Byte	SRQ on Numeric Entry Completed (HP–IB or Front Panel)	SRQ on Any Front Panel Key Pressed
EXTENDED STATUS BYTE (#2)								
BIT #	7	6	5	4	3	2	1	0
Decimal Value	128	64	32	16	8	4	2	1
Function	N/A	SRQ on Detector Uncal	SRQ on Front Panel Preset or Power–on	SRQ on Limit Test Failed[1]	SRQ on Action Requested not possible	SRQ on Knob Activity	SRQ on Operation Failed[1]	SRQ on Self Test Failure

1. HP 8757D only.

To use the sweep hold mode, modify line 100 to "SW2" and delete lines 120 and 130. The program will wait at line 140 until the 10 sweeps are completed. Whenever practical, use the service request interrupt to sense the end of a TAKE SWEEP command. In fact, you can use the time to do plotting or printing of data, instead of sitting in a loop. Service requests are also useful for other events, as demonstrated by the next program.

Program 8: programming the softkeys

The analyzer has eight screen–labeled softkeys that make your measurements faster and easier. Under HP–IB control, you can re–label the softkeys with any annotation and sense when they are pressed.

Use the softkeys to branch to special measurement programs. By making full use of the softkeys, your automatic system may not need a normal computer keyboard at all, making it as easy to use as a manual instrument.

Program 8 listing

```
10    PRINTER IS 1
20    ASSIGN @Sna TO 716
30    ABORT 7
40    CLEAR @Sna
50    OUTPUT @Sna;"IP"
60    OUTPUT @Sna;"CS RM8"
70    ON INTR 7 GOTO Srq_recv
80    OUTPUT @Sna;"WK1 CAL 1"
90    OUTPUT @Sna;"WK2 TEST 1"
100   OUTPUT @Sna;"WK3 CAL 2"
110   OUTPUT @Sna;"WK4 TEST 2"
120   OUTPUT @Sna;"WK8 ABORT"
130   Wait_srq: !
140     ENABLE INTR 7;2
150     GOTO 150
160   Srq_recv: !
170     Stat=SPOLL(@Sna)
180     OUTPUT @Sna;"OK"
190     ENTER @Sna;Key code
200     SELECT Key code
210       CASE =32
220         PRINT "Calibration #1"
230       CASE =8
240         PRINT "Test #1"
250       CASE =0
260         PRINT "Calibration #2"
270       CASE =16
280         PRINT "Test #2"
290       CASE =41
300         PRINT "Abort measurement"
310         GOTO 360
320       CASE ELSE
330         PRINT "*** undefined ***"
340     END SELECT
350     GOTO Wait_srq
360     END
```

Program 8 explanation

Line 10 Direct output to the CRT of the computer.

Line 20 Assign an I/O channel to the address of the analyzer.

Line 30 Abort any transfers and clear the HP–IB interface of the computer.

Line 40 Clear the HP–IB interface of the analyzer.

Line 50 Preset the analyzer and source.

Line 60 Set the request mask to interrupt the computer whenever a softkey is pressed (bit 3). See Table 1 for the description of the status bytes.

Line 70 Define the line that the computer will go to whenever it receives an interrupt.

Line 80 Label softkey 1 with "CAL 1". Softkey 1 is the softkey at the top of the CRT.

Line 90 Label softkey 2 with "TEST 1".

Line 100 Label softkey 3 with "CAL 2".

Line 110 Label softkey 4 with "TEST 2".

Line 120 Label softkey 8 with "ABORT".

Line 130 Line label for routine that waits for an interrupt.

Line 140 Turn on the SRQ interrupts in the computer.

Line 150 Wait for the interrupt in a tight loop. If PAUSE were used, the interrupts would not be active.

Line 160 Line label for the routine that services the interrupts.

Line 170 Serial poll the analyzer. Reading the status byte of the analyzer clears the SRQ. The CLEAR STATUS (CS) command could also be used.

Line 180 Command the analyzer to output the key code of the last key pressed.

Line 190 Read the key code.

Line 200 Multi–way branch on key code value.

Line 210 If the key code is 32, then softkey 1 was pressed.

Line 220 Print an appropriate message.

Line 230 If the key code is 8, then softkey 2 was pressed.

Line 240 Print an appropriate message.

Line 250 If the key code is 0, then softkey 3 was pressed.

Line 260 Print an appropriate message.

Line 270 If the key code is 16, then softkey 4 was pressed.

Line 280 Print an appropriate message.

Line 290 If the key code is 41, then softkey 8 was pressed.

Line 300 Print an appropriate message.

Line 310 Exit the program by jumping to the end.

Line 320 If the key code doesn't match any of the preceding codes, another key was pressed. In this case, the key code has to be for softkey 5, 6, or 7 (key codes 14, 38, or 40) since these are the only other keys that can interrupt the computer.

Line 330 Print an appropriate message.

Line 340 End of multi–way branch.

Line 350 Re–enter the program at the "Wait_srq" label. At that point, the interrupts are re–enabled and the computer waits for another SRQ.

Line 360 End program execution.

Running program 8

1. Clear the program memory of the computer and type in the program.
2. Press **[RUN]** on the computer.
3. After the computer presets the analyzer and source, it writes the softkey labels on the analyzer CRT. When the first key label is written, the analyzer labels it and blanks the other softkey labels. Since all labels except softkeys 5, 6, and 7 are given new labels, softkeys 5, 6, and 7 remain blank.
4. Press any key on the analyzer except the ***[ABORT]*** softkey. Pressing a softkey causes a message to be printed on the CRT of the computer. Note that softkeys 5, 6, and 7 generate an interrupt, even though they weren't labeled. No other keys of the analyzer generate an interrupt, because of the SRQ mask specified. Because the analyzer is in remote mode, nothing is changed by pressing its keys.
5. Press the ***[ABORT]*** softkey to end program execution.

In this example, the service request mask was set to interrupt the computer whenever a softkey was pressed. Another bit in the mask causes an interrupt to be generated when any key is pressed.

Because the analyzer was left in remote mode, it didn't respond to any keys pressed on its front panel. In some applications it is useful to put the analyzer into local operation, so that it can be controlled from the front panel and still generate interrupts whenever a key is pressed.

Program 9: CRT graphics

For applications requiring diagrams, drawings, or limit lines, the CRT of the analyzer may be written to as if it were a Hewlett–Packard plotter. By defining the analyzer as the plot device used by the computer, you can even use the special plotting statements built into the computer, such as MOVE, DRAW, PEN, AXES, VIEWPORT, etc.

This program draws a connections diagram for a hypothetical test system measuring an amplifier. It will blank the analyzer's standard display containing the graticule, annotation, and softkeys so that we have a blank CRT. Figure 2 shows what the CRT should look like when the program is done. Since the program involves drawing many lines, it will use the BASIC data statement to more efficiently store where to draw lines.

For fast, easy–to–use graphics, the graphics memory of the HP 8757D/E is divided into seven "pages" of 500 words and an eighth page of 4000 words. One vector requires two words. Each of the pages may be selected to receive data, and be turned on and off independently. You can keep different drawings in each of the graphics memory pages and simply turn on the drawing you need by turning on the appropriate page. Each page may also be erased independently.

To use the graphics capability of the HP 8757D/E, first define the passthru address to be one digit less than the analyzer's control address. If the analyzer's address is 16, its graphics address is 15. To the computer, the CRT of the analyzer looks like a plotter connected to the 8757 SYSTEM INTERFACE.



Figure 2. The CRT Graphics Display

Program 9 listing

```
10   ASSIGN @Sna TO 716
20   ASSIGN @Passthru TO 717
30   ABORT 7
40   CLEAR @Sna
50   OUTPUT @Sna;"IP BL5 PT15"
60   GINIT
70   PLOTTER IS 717,"HPGL"
80   WINDOW 0,2924,0,2047
90   CLIP 0,2900,0,2000
100  OUTPUT @Passthru;"EP; GP1,1; DF"
110  PEN 3
120  GRID 100,100
130  PEN 10
140  RESTORE Graphix
150  REPEAT
160    READ Pen_mode$,X,Y
170    SELECT Pen_mode$
180      CASE "D"
190        DRAW X,Y
200      CASE "M"
210        MOVE X,Y
220    END SELECT
230  UNTIL Pen_mode$="E"
240  MOVE 600,1600
250  OUTPUT @Passthru USING "K";"SI0.28,0.34;
     LBCONNECTION DIAGRAM";CHR$(3)
260  MOVE 1200,250
270  LABEL "DUT"
280 Graphix: !
290  DATA "M",300,800,"D",1100,800,"D",1100,1100,
     "D",300,1100
300  DATA "D",300,800,"M",800,800,"D",800,1100
310  DATA "M",1500,800,"D",2300,800,"D",2300,1200,
     "D",1500,1200
320  DATA "D",1500,800,"M",1950,800, "D",1950,1200
330  DATA "M",875,850,"D",875,500,"D",1200,500
340  DATA "M",1400,500,"D",2050,500,"D",2050,850
350  DATA "M",1200,400,"D",1400,500,"D",1200,
     600,"D",1200,400
360  DATA "E",0,0
370  END
```

Program 9 explanation

Line 10 Assign an I/O path to the address of the analyzer.

Line 20 Assign an I/O path to the passthru address of the analyzer.

Line 30 Abort any transfers and clear the HP–IB interface of the computer.

Line 40 Clear the HP–IB interface of the analyzer.

Line 50 Preset the analyzer and blank the CRT display. Define the CRT graphics as the target of passthru commands. The graphics address is always one less than the analyzer's HP–IB address.

Line 60 Initialize the graphics. This sets a default line type, scale, and clipping limits in the computer.

Line 70 Define the analyzer CRT as the plot device and tell the computer that it is an HP–GL (Hewlett–Packard Graphics Language) device.

Line 80 Scale the plotting area to the entire CRT. The numbers are the corners of the CRT (the CRT is described in the *Operating Reference*.)

Line 90 Define the soft clip area to maintain a clean display.

Line 100 Erase all graphics pages. Turn on graphics page 1 to ensure that the graphics start in it. Set the display to monochrome default colors.

Line 110 Select to plot with pen 3, dim green.

Line 120 Plot a grid on the CRT. These are 100 by 100 graphic units per square, giving you an indication of where the X and Y coordinates are on the CRT.

Line 130 Select to plot with pen 10, the brightest intensity for the analyzer CRT.

Line 140 Define where to start looking for data. Here we've indicated that the data starts at the line label "Graphix", which is line 280. This ensures that we always start at the right data statement.

Line 150 Define the beginning of a loop.

Line 160 Read three items from the data statement. "Pen_mode$" is a one character string indicating whether we should move (M), draw (D), or end (E) the plotting. X and Y are the plotting coordinates.

Line 170 Multi–way branch on the "Pen_mode$" value.

Line 180 If "Pen_mode$" is "D", then we want to draw.

Line 190 Draw to coordinates X,Y.

Line 200 If "Pen_mode$" is "M", then we want to move.

Line 210 Move to coordinates X,Y.

Line 220 End of multi–way branch.

Line 230 End of the repeat loop. Repeat lines 160 through 220 again if "Pen_mode$" isn't "E". If it was, then we are done plotting the data in the data statements.

Line 240 Move the pen to title our display.

Line 250 Title the display with the label "CONNECTION DIAGRAM". This shows one way to label the analyzer display by using its internal character set. To do this, we must first specify which set to use via the SI command. This specifies the width and height respectively of each character and is similar to the computer's CSIZE instruction. We indicate what the label is with the LB command and follow it with the label. We must terminate the label with an "end–of–text" (ETX) character (a byte equal to a binary 3.)

Line 260 Move the pen to label our device under test (DUT).

Line 270 Label the DUT using the computer's LABEL statement. Notice the difference between this label and the one generated in line 250. First, the intensity is less. Second, the characters look more round and smooth. This is because the computer generates each character by plotting several small strokes (more than the HP 8757D/E's internal CRT does for its characters.) This means that you will also use much more graphics memory than with the internal character set.

Line 280 Define the start of the data statements containing our plotting information for all of the lines on the CRT. While these may be less legible than lots of MOVEs and DRAWs, it is more efficient programming.

Line 290 This data statement draws the outline of the source.

Line 300 This data statement draws the RF plug–in.

Line 310 This data statement draws the outline of the analyzer.

Line 320 This data statement draws the CRT of the analyzer.

Line 330 This data statement draws the connections from the source to the DUT.

Line 340 This data statement draws the connections from the DUT to the analyzer.

Line 350 This data statement draws the DUT (an amplifier.)

Line 360 This data statement indicates the end of our plotting. The X and Y values are needed here only for the read statement in line 160.

Line 370 End program execution.

Running program 9

1. Clear the program memory of the computer and type in the program.
2. Press **[RUN]** on the computer.
3. After the analyzer and source are preset, the CRT will be blanked. First a grid is plotted on the CRT. While this isn't necessary for our connection diagram, it does give you a good indication of where the X and Y coordinates are on the analyzer's CRT.
4. All of the lines are plotted on the analyzer's CRT. These are just a sequence of MOVEs and DRAWs as specified by the data statements. If brighter lines are desired, draw each line twice.
5. Finally the labeling is added. The label "CONNECTION DIAGRAM" is done using the analyzer CRT's internal character set. The "DUT" label uses the computer's character set. The significant differences are that while using the computer's LABEL statement is easier, it also takes a lot more graphics memory than the internal character set. This can become very important is you have several labels or want to have several hookup diagrams.

In this example, only graphics page 1 is used. You can independently control up to 7 separate pages of graphics information. If you write too much information into one page, it overflows into the next page.

When a graphics page is selected, the first location of memory that receives information (identified by the "pointer") is reset to the beginning of the page. Thus, as information is written into the page, the old information is destroyed. If we were plotting a line, this would appear as a new trace overwriting an old one.

Program 10: learning the instrument state

Being able to save a specific instrument state is helpful when it is needed several times in a test or measurement procedure. You can save the instrument state by manually logging the important analyzer and sweeper parameters, such as start/stop frequency, sweep time, number of trace points, scale per division, and display format, then re−input them at the appropriate time. A simpler approach is to save the instrument state in one of the nine internal save/recall registers of the analyzer/source combination, then recall it when needed.

You have two additional options with HP−IB: the interrogate function and the learn string. With the output interrogated parameter function (OP), you can selectively interrogate the values of all functions that have numeric values (such as frequency and number of trace points.) This function operates the same in both the analyzer and the source. It is illustrated in program 3 where the source start and stop frequencies are interrogated in lines 80 through 130.

For a more thorough approach, use the learn string functions of the analyzer and source. Learn string describes the present instrument state and is similar to one of the internal save/recall registers. For the analyzer, the learn string also includes all of the global parameters, but does not include limit line information. Once an instrument state is learned, the analyzer and source states can be restored at any later time. The following program demonstrates how to both learn and restore the instrument states of the HP 8757Dand the HP 8350B Sweep Oscillator by using their learn string functions. If you use the HP 8757E or the HP 8340B or 8341B Synthesized Sweepers, perform the modification described at the end of "Program 10 Explanation".

Program 10 listing

```
10     OPTION BASE 1
20     DIM Lswpr$[90],Lsna$[150]
30     ASSIGN @Sna TO 716
40     ASSIGN @Passthru TO 717
50     ABORT 7
60     CLEAR @Sna
70     OUTPUT @Sna;"PT19;"
80     LOCAL @Sna
90     INPUT "SET UP SYSTEM, PRESS CONTINUE",A$
100    OUTPUT @Sna;"OL"
110    ENTER @Sna USING "#,300A";Lsna$
120    OUTPUT @Passthru;"OL"
130    ENTER @Passthru USING "#,90A";Lswpr$
140    OUTPUT @Sna;"IP"
150    INPUT "TO RESTORE SETUP, PRESS CON-
       TINUE",A$
160    OUTPUT @Sna USING "2A,300A";"IL",Lsna$
170    OUTPUT @Passthru USING "2A,90A";"IL",Lswpr$
180    OUTPUT @Sna
190    LOCAL @Sna
200    END
```

Program 10 explanation

Line 10	Define the first element of any array to be at index number 1.
Line 20	Dimension two strings large enough to hold the learn strings of the source (90 bytes) and the analyzer (300 bytes).
Line 30	Assign an I/O path to the address of the analyzer.
Line 40	Assign an I/O path to the passthru address of the analyzer.
Line 50	Abort any transfers and clear the HP−IB interface of the computer.
Line 60	Clear the HP−IB interface of the analyzer.
Line 70	Tell the analyzer which device is controlled through the passthru address. Address 19 belongs to the source.
Line 80	Set the analyzer and source to local mode.

Line 90 Prompt the user to set up the system and wait for the Continue key press.

Line 100 Program the analyzer to output its learn string.

Line 110 Read the analyzer learn string into the string "Lsna$". Notice the "#,150A" format. The HP 8757D learn string is 150 contiguous binary bytes that does not end with a cr/lf (since these could actually be part of the learn string information). The computer must read all 300 bytes and this format ensures that it will.

Line 120 Program the source to output its learn string.

Line 130 Read the source learn string into the string "Lswpr$". Notice the "#,90A" format. As on line 110, the computer must read the entire source learn string. For the HP 8350B Sweep Oscillator, it is 90 bytes long.

Line 140 Preset the analyzer and source to clear the instrument states.

Line 150 Prompt the user and wait for the [Continue] key press.

Line 160 Program the analyzer to accept its learn string, then send the learn string. Notice the "2A,300A" format ensures that the IL command and the 300 bytes of the learn string are sent continuously. The HP 8757D expects the learn string to start immediately after the IL command.

Line 170 Program the source to accept its learn string, then send the learn string. Notice the "2A,90A" format. As on line 160, this ensures that the source learn string is sent properly.

Line 180 Re–address the analyzer to exit passthru mode and continue sweeping.

Line 190 Set the analyzer and source to local mode.

Line 200 End of execution.

Running program 10

1. Clear the program memory of the computer and type in the program.
2. Press **[RUN]** on the computer.
3. The computer stops and displays: `SET UP SYSTEM, PRESS CONTINUE.`

   Adjust the analyzer and source to a preferred instrument state, then press **[Continue]** on the computer.
4. The computer will save the learn strings of both the analyzer and the source. After completing this, the analyzer and source will be preset to destroy your original instrument state.
5. The computer stops and displays: `TO RESTORE SETUP, PRESS CONTINUE.`

Press **[Continue]** on the computer. The computer will restore your original instrument state via the two learn strings. Verify on the displays of the analyzer and the source that your instrument state has been restored.

This example is designed to work with the HP 8757D, which has a learn string of 300 bytes, and the HP 8350B Sweep Oscillator, which has a learn string of 90 bytes. To modify the program to work with the HP 8757E, which has a learn string of 150 bytes, change the "300" in lines 20, 110, and 160 to "150". To modify the program to work with the HP 8340B and 8341B Synthesized Sweepers, which have learn strings 123 bytes in length, change the "90" in lines 20, 130, and 170 to "123".

Program 11: CRT graphics on the HP 8757D

As was illustrated by program 9, it is possible to utilize the CRT of the HP 8757D/E as a plotter. This program goes one step further by utilizing the CRT to create a simple connection diagram which can be recalled by the user, at any time, from the front panel of the analyzer.

This program draws the same hypothetical connection diagram as was drawn by program 9. It blanks most of the analyzer's standard display including the graticle and all annotation except the softkeys. In addition it adds one softkey under both the save and the recall hardkey menus. This softkey will allow the user to toggle the state of the CRT graphics off and on.

To use the graphics off/on capability of the analyzer, simply change "BL5" in line 50 of program 9 to "BLA" and make the necessary changes in the size of the background grid. These changes are illustrated in the following listing.

The same principle may be used to save to disk anything stored in the first seven pages of user graphics on the analyzer. By having the softkeys available, the user can store CRT graphics onto a disk for later recall.

Program 11 listing

```
10      ASSIGN @SNA TO 716
20      ASSIGN @PASSTHRU to 717
30      ABORT 7
40      CLEAR @Sna
50      OUTPUT @Sna;"IP BLA PT15"
60      GINIT
70      PLOTTER IS 717,"HPGL"
80      WINDOW 0,2924,0,2047
90      CLIP 0,2700,0,2000
100     OUTPUT @Passthru;"EP; GP1,1;DF"
110     PEN 3
120     GRID 100,100
130     PEN 10
140     RESTORE Graphix
150     REPEAT
160        READ Pen_mode$,X,Y
170        SELECT Pen_mode$
```

```
180            CASE "D"
190              DRAW X,Y
200            CASE "M"
210              MOVE X,Y
220         END SELECT
230       UNTIL Pen_mode$="E"
240       MOVE 600,1600
250       OUTPUT @Passthru USING
          "K";"SI0.28,0.34;LBCONNECTION DIA-
          GRAM";CHR$(3)
260       MOVE 1200,250
270       LABEL "DUT"
275       OUTPUT@Sna
280       LOCAL 7
290   Graphix: !
300       DATA
          "M",300,800,"D",1100,800,"D",1100,1100,"D",300,1100
310       DATA "D",300,800,"M",800,800,"D",800,1100
320       DATA
          "M",1500,800,"D",2300,800,"D",2300,1200,"D",1500,12
          00
330       DATA "D",1500,800,"M",1950,800,"D",1950,1200
340       DATA "M",875,850,"D",875,500,"D",1200,500
350       DATA "M",1400,500,"D",2050,500,"D",2050,850
360       DATA
          "M",1200,400,"D",1400,500,"D",1200,600,"D",1200,400
370       DATA "E",0,0
380       END
```

Program 11 explanation

Line 10 Assign an I/O path to the address of the analyzer.

Line 20 Assign an I/O path to the passthru address of the analyzer.

Line 30 Abort any transfers and clear the HP–IB interface of the computer.

Line 40 Clear the HP–IB interface of the analyzer.

Line 50 Preset the analyzer and blank all the CRT display except the softkeys. Define the CRT graphics as the target of passthru commands. The graphics address is always one less than the analyzer's HP–IB address.

Line 60 Initialize the graphics. This sets a default line type, scale, and clipping limits in the computer.

Line 70 Define the analyzer CRT screen as the plot device and tell the computer that it is an HP–GL (Hewlett–Packard Graphics Language) device.

Line 80 Scale the plotting area to the CRT screen, allowing space for the softkeys. The numbers are the corners of the CRT, as described in the analyzer's operating manual.

Line 90 Define the soft clip area to maintain a clean display.

Line 100 Erase all graphics pages. Turn on graphics page 1 on to ensure that the graphics start there.

Line 110 Select to plot with pen 3, the lowest intensity for the analyzer CRT.

Line 120 Plot a grid on the CRT. These are 100 by 100 squares, giving you an indication of where the X and Y coordinates are on the CRT.

Line 130 Select to plot with pen 1, the brightest intensity for the analyzer CRT.

Line 140 Define where to start looking for data. Here we've indicated that the data starts at the line label "Graphix", which is line 280. This ensures that we always start at the right data statement.

Line 150 Define the beginning of a loop.

Line 160 Read three items from the data statement. Pen_mode$ is a one character string indicating whether we should move "M", draw "D", or end "E" the plotting. X and Y are the plot coordinates.

Line 170 Multi–way branch on the Pen_mode$ value.

Line 180 If Pen_mode$ is "D", then we want to draw.

Line 190 Draw to coordinates X,Y.

Line 200 If Pen_mode$ is "M", then we want to move.

Line 210 Move to coordinates X,Y.

Line 220 End of multi–way branch.

Line 230 End of the repeat loop. Repeat lines 160 through 220 again if Pen_mode$ isn't "E". If it was, then we are finished plotting the data in the data statements.

Line 240 Move the pen to title our display.

Line 250 Title the display with the label "CONNECTION DIAGRAM". This shows one way to label the analyzer display by using its internal character set. To do this, we must first specify which set to use via the "SI" command. This specifies the width and height respectively of each character and is similar to the computer's CSIZE instruction. We indicate what the label is with the "LB" command and follow it with the label. We must terminate the label with an "end–of–text" (ETX) character –a byte equal to a binary 3.

Line 260 Move the pen to label our device under test (DUT).

Line 270 Label the DUT using the computer's LABEL statement. Notice the difference between this label and the one generated in line 250. First, the intensity is less. Second, the characters look more round and smooth. This is because the computer generates each character by plotting several small strokes (more than the HP 8757D/E's internal CRT does for its characters). This means that you will also use much more graphics memory than with the internal character set.

Line 275 Exit from passthrough mode.

Line 280 Set the analyzer and the source to local mode.

Line 290 Define the start of the data statements containing our plotting information for all of the lines on the CRT. While these may be less legible than lots of MOVEs and DRAWs, it is more efficient programming.

Line 300 This data statement draws the outline of the sweeper.

Line 310 This data statement draws the plug–in in the sweeper.

Line 320 This data statement draws the outline of the analyzer.

Line 330 This data statement draws the CRT of the analyzer.

Line 340 This data statement draws the connections from the sweeper to the DUT.

Line 350 This data statement draws the connections from the DUT to the analyzer.

Line 360 This data statement draws the DUT (an amplifier.)

Line 370 This data statement indicates the end of our plotting. The X and Y values are needed here only to keep the read statement in line 160 happy.

Line 380 End program execution.

Running program 11

1. Clear the program memory of the computer and type in the program.
2. Press **[RUN]** on the computer.
3. After the analyzer and source are preset, the CRT will be blanked except the softkeys. First a grid is plotted on the CRT. While this isn't necessary for our connection diagram, it does give you a good indication of where the X and Y coordinates are on the analyzer's CRT.
4. All of the lines are plotted on the analyzer's CRT. These are just a sequence of MOVEs and DRAWs as specified by the data statements. If brighter lines are desired, draw each line twice, or select different pen numbers.
5. Finally the labeling is added. The label "CONNECTION DIAGRAM" is done using the analyzer CRT's internal character set. The "DUT" label was done using the computer's character set. The key differences are that while using the computer's LABEL statement is easier, it also takes a lot more graphics memory than the internal character set. This can become very important if you have several labels or want to have several hookup diagrams.

Program 12: reading disks from the HP 8757D

In many cases it may be necessary to manipulate data that was saved onto disk with the HP 8757D. This program can be used to read the data files and display the contents of those files on the computer's CRT display. No frequency information is read or displayed.

The CITIfile (Common Instrumentation Transfer and Interchange File) disk format is used on the data disks for the HP 8757D. This program reads in a single array of data from a CITIfile data disk. It then outputs the point number and the magnitude associated with that point.

In order to use this program you must have saved either memory or measurement data in files onto a disk with the HP 8757D. In addition this data needs to be stored in ASCII format in order for it to be read by the CITIfile routine.

Program 12 listing

```
10    INTEGER Intvar(1:30),Counter
20    REAL Data_pt(1:1601,1:2,1:1)
30    DIM Filename$[30]
40    LINPUT "Name of file to read? ",Filename$
50    Read_citifile(Filename$,Data_pt(*),Intvar(*))
60    PRINT "POINT #REAL IMAG "
70    FOR Counter=1 TO Intvar(1)
80      PRINT Counter;TAB(10);Data_pt(Counter,1,1);
        TAB(24);Data_pt(Counter,2,1)
90      NEXT Counter
100   PRINT "Printed the data array from file ";Filename$;"."
110   PRINT "The file contained an array with ";Intvar(1);"
      data points."
120   END
130   SUB Read_citifile(Filename$,Data_pt(*),INTEGER
      Intvar(*))
140   INTEGER Done,Count1
150   ALLOCATE Current_line$[256],Token_found$[32]
160   ASSIGN @Disk TO Filename$
170   Done=0
180   REPEAT
190     ENTER @Disk;Current_line$
200     Current_line$=TRIM$(Current_line$)
```

```
210        GOSUB Get_next_token
220        IF (Token_found$<>" ") THEN
230          SELECT Token_found$
240          CASE "VAR"
250            FOR Count1=1 TO 2
260              GOSUB Remove_one_word
270            NEXT Count1
280            Intvar(1)=VAL(Current_line$)
290          CASE "BEGIN"
300            ALLOCATE Dstring$(1:Intvar(1))[30]
310            ENTER @Disk;Dstring$(*)
320            FOR Count1=1 TO Intvar(1)
330              Data_pt(Count1,1,1)=VAL(Dstring$(Count1))
340              Data_pt(Count1,2,1)=VAL(Dstring$(Count1)
                 [POS(Dstring$(Count1),",")+1])
350            NEXT Count1
360            Done=1
370          END SELECT
380        END IF
390      UNTIL Done
400      SUBEXIT
410    Get_next_token: !
420      Space_pos=POS(Current_line$," ")
430      IF Space_pos=0 THEN
440        Token_found$=Current_line$
450        Current_line$=" "
460      ELSE
470        Token_found$=Current_line$[1,Space_pos-1]
480        GOSUB Remove one word
490      END IF
500      RETURN
510    Remove_one_word:Current_line$=TRIM$
       (Current_line$[POS(Current_line$," ")+1])
520      RETURN
530      SUBEND
```

Program 12 Explanation

Line 10 Declare an integer variable and dimension an integer array for use within the program.

Line 20 Dimension a real array for data storage.

Line 30 Dimension and reserve memory for the filename string.

Line 40 Prompt for and, accept alphanumeric input from the keyboard and place it in the "Filename$" string.

Line 50 Call the subprogram that reads the file named via line 40.

Line 60 Setup a header for the printout of data from the array read in by the "Read citifile" subprogram.

Line 70 Set up a counter to determine the number of points to be printed from the data array. Note that the length of the array is kept in "Intvar(1)". This value is returned from the "Read_citifile" subprogram call.

Line 80 Print the data point, the REAL component, and the IMAGINARY component. For a scalar analyzer the imaginary components are always presumed to be zero. It is illustrative here to point out that the most general means of storing data in all analyzers requires the inclusion of phase as well as magnitude data. This is accomplished by storing the linear REAL and IMAGINARY components of the measurements.

Line 90 Continue printing until the entire data array is completed.

Line 100 Print the name of the file just listed at the end of the listing.

Line 110 Finish the printout by printing the number of data points in the data array.

Line 120 End of the main program.

Line 130 Define a subprogram that can be used to input the data array for the file named by the user. This routine is written to be modular so that it can be easily adapted for use in other programs.

Line 140 Declare the INTEGER variables to be used in the subprogram.

Line 150 Dynamically allocate memory for string variables used in the subprogram.

Line 160 Assign an I/O path name to the mass storage file of interest. In this case it is the filename input by the user.

Line 170 Set the expression "Done" to zero. This will be used to determine if all data has been read into the data array.

Line 180 Define a loop which is repeated until the value for "Done" is greater than zero.

Line 190 Statement used to input data from the file of interest and assign the values entered to a string variable.

Line 200 This function is used to return the string stripped of all leading and trailing ASCII spaces.

Line 210 Call a subprogram used to locate the first word on the line.

Line 220 If the boolean expression within the parenthesis is evaluated as true then the following conditions are tested. In this case the expression can be set true by the "Get_next_token" subprogram.

Line 230 Provide for conditional execution of one of two program statements using the string expression.

Line 240 If the string expression is "VAR", then remove two words from the data array. This is done to locate a piece of data that specifies the length of the array to follow.

Line 250 Begin loop to remove two words.

Line 260 Call to subprogram which removes a word.

Line 270 Continue the loop until two words have been removed.

Line 280 Convert the string expression for the length of the data array into a numeric value that can be used in the program as a counter.

Line 290 If the string expression is "BEGIN" this signifies the beginning of the data array to be input.

Line 300 Dynamically allocate space for the data array to be input.

Line 310 Input the data from the file specified.

Line 320 Begin a loop which converts the data string to numeric data pairs.

Line 330 Convert a REAL data point from the string into a numeric value and store the numeric value in an array.

Line 340 Convert an IMAGINARY data point from the string into a numeric value. The data is listed in pairs, separated by a comma, with the REAL component preceding the IMAGINARY component.

Line 350 Proceed to the next data pair.

Line 360 After completing input of the data array, set "Done" greater than 0.

Line 370 Finish of the construct which allowed for the conditional execution of one of two cases.

Line 380 Complete IF .... THEN sequence.

Line 390 Check to see that "Done" is greater than 0.

Line 400 This statement is used to return from the subprogram at some point other than the SUBEND statement. It allows for more than one exit from a subprogram.

Line 410 Define a subprogram called by "Read_citifile" to locate the next "token". The tokens of interest to this program are "VAR" and "BEGIN".

Line 420 Locate the position of the next ASCII space within the data string.

Line 430 If the value returned for the position of the next ASCII space is 0, then the ASCII space character doesn't exist in the string being searched.

Line 440 Set the token to whatever is currently in the data string.

Line 450 Set the string to a null character.

Line 460 If the value returned is greater than zero, then perform the following operations.

Line 470 Remove keyword from the string and make it the current token.

Line 480 Increment the pointer along the data string.

Line 490 Complete IF ... THEN sequence.

Line 500 Return from the "Get_next_token" subprogram.

Line 510 Define a subprogram called by "Get_next_token" subprogram. Used to increment the pointer along the data string to the next word.

Line 520 Return from the "Get_next_token" subprogram.

Line 530 End of the "Read_citifile" subprogram.

Running Program 12

1. Clear the program memory of the computer and type in the program.
2. Press **[RUN]** on the computer.
3. The computer displays the prompt:

   `Name of file to read?`

   At the prompt enter the name of the ASCII data file to be read by the program then press **[Continue]**. The program will begin reading the file specified and display the data from the disk onto the CRT of the computer in a tabular format.
4. The data will be listed as linear values for both the real and the imaginary components.

Table 2. Alphabetical Listing of HP 8757D/E Programming Codes (1 of 3)

Code	Action	Code	Action
A0	Averaging off	CLS	Color list, salmon[1]
AB	A/B ratio measurement	CLW	Color list, white[1]
AC	A/C ratio measurement[2]	CLY	Color list, yellow[1]
AFd	Averaging on and factor d	CN	Cursor to minimum
ANm	Adaptive Normalization on/off	COBd	Brightness adjust, one color[1]
AR	A/R ratio measurement	COCd	Color adjust, one color[1]
AS	Autoscale	COTd	Tint adjust, one color[1]
AZ2	Autozero the DC detectors once	CR	C/R ratio measurement[2]
AZm	Autozero repeat on/off of the DC detectors	CS	Clear status bytes
BA	B/A ratio measurement	CTm	Auto system calibration on/off
BC	B/C ratio measurement[2]	CUm	Cursor on/off
BFm	Plotter buffer on/off[3]	CWm	CW mode on/off
BL0	Restore CRT to normal mode	CX	Cursor to maximum
BL1	Blank frequency labels (secure frequency mode, frequency labels cannot be restored)	DAd	Detector A amplitude offset set to d
		DBd	Detector B amplitude offset set to d
BL2	Blank all labels	DCd	Detector C amplitude offset set to d[2]
BL3	Blank active channel trace	DEC	Set default colors[1]
BL4	Blank softkey labels	DFA	Set disk format to ASCII[1]
BL5	Blank all (except user CRT graphics)	DFB	Set disk format to binary[1]
BL6	Blank title	DFE	Set Disk format to extended binary[1]
BL7	Blank mode labels	DHm	Display Hold on/off of the active channel trace
BL8	Blank the active entry area		
BL9	Blank the limit lines	DIAd	Set disk HP−IB address[1]
BLA	Blank all (except user CRT graphics and softkeys)	DIUd	Set disk unit number[1]
		DIVd	Set disk volume number[1]
BR	B/R ratio measurement	DLF	Delete file from disk[1]
BTNd	Overall display brightness	DM0	All inputs set to DC detection
BW	Display the search bandwidth on the CRT[1]	DM1	All inputs set to AC detection
C0	Channel off	DN	Step down (decrement)
C1	Channel 1 on/active	DOAd	Measure Detector A amplitude offset
C2	Channel 2 on/active	DOBd	Measure Detector B amplitude offset
C3	Channel 3 on/active[1]	DOCd	Measure Detector C amplitude offset[2]
C4	Channel 4 on/active[1]	DORd	Measure Detector R amplitude offset
CA	C/A ratio measurement[2]	DRd	Detector R amplitude offset set to d
CB	C/B ratio measurement[2]	DS0	Display trace data in log magnitude
CC1	Set channel 1 color[1]	DS1	Display trace data in standing wave ratio (SWR) format
CC2	Set channel 2 color[1]		
CC3	Set channel 3 color[1]	DTSTPAs	Enter stop frequency for detector A
CC4	Set channel 4 color[1]	DTSTPBs	Enter stop frequency for detector B
CDm	Cursor delta on/off	DTSTPCs	Enter stop frequency for detector C[2]
CGL	Set labels color[1]	DTSTPRs	Enter stop frequency for detector R
CGN	Set background color[1]	DTSTRAs	Enter start frequency for detector A
CGR	Set grid color[1]	DTSTRBs	Enter start frequency for detector B
CGW	Set warning label color[1]	DTSTRCs	Enter start frequency for detector C[2]
CL	Perform system configuration of detectors and channels	DTSTRRs	Enter start frequency for detector R
		EO	Enter measured detector amplitude offset
CLB	Color list, black[1]	ER0	Erase all save/recall registers
CLG	Color list, green[1]	FAs	Start frequency label
CLL	Color list, blue[1]	FBs	Stop frequency label
CLR	Color list, red[1]	FD0	Format data ASCII

1. HP 8757D only
2. HP 8757D Option 001 only
3. Revision 3.1 or above for HP 8757E.

Table 2. Alphabetical Listing of HP 8757D/E Programming Codes (2 of 3)

Code	Action
FD1	Format data binary (HP BASIC compatible)
FD2	Format data extended ASCII
FD3	Format data binary (PC compatible)
FD4	Format data extended binary (HP BASIC compatible)
FD5	Format data extended binary (PC compatible)
FR0	Logarithmic (dB) cursor format[3]
FR1	SWR cursor format[3]
FSm	Step sweep on/off[3,4]
FTAm	Detector A frequency on/off
FTBm	Detector B frequency on/off
FTCm	Detector C frequency on/off[2]
FTRm	Detector R frequency on/off
IA	Input A absolute power measurement
IB	Input B absolute power measurement
IC	Input C absolute power measurement[2]
ILs	Input Learn string
IND	Initialize disk format[1]
IP	Instrument preset
IR	Input R absolute power measurement
IX	External ADC input (AUX) voltage measurement1
LE	Erase limit lines for active channel[5]
LFA	Load instrument information file from disk[1]
LFC	Load CRT graphics file from disk[1]
LFD	Load data trace file from disk[1]
LFF	Load measurement file from disk.[1]
LFH	Load instrument information file from disk and place instrument in hold mode.[1]
LFI	Load instrument state file from disk[1]
LFM	Load memory trace file from disk[1]
LFN	Load display trace file from disk.[1]
LFs	Enter limit test flat line data[5]
LL	Store lower limit line into memory[5]
LPs	Enter limit test point data[5]
LSs	Enter limit test sloped line data[5]
LTm	Limit line test on/off[5]
LU	Store upper limit line into memory[5]
M−	Display normalized data (measurement − memory)
MDm	Modulation on/off
ME	Display measurement data
MM	Display the channel menu(main menu)
MN	Display normalized data (same as M−)
MOC	Monochrome display[1]
MR	Marker (or cursor) to reference line
MSm	Manual sweep mode on/off
MU0	Display the measurement menu
MU1	Display the display menu
MU2	Display the scale menu

Code	Action
MU3	Display the reference menu
MU4	Display the cursor menu
MU5	Display the average menu
MU6	Display the calibration menu
MU7	Display the special menu
MU8	Display the system menu
MY	Display memory data
MZ	Manual calibration of DC detectors
NSm	Non−standard sweep mode on/off
OC	Output cursor value
OD	Output trace data
OE1	Output error status of display channel 1
OE2	Output error status of display channel 2
OI	Output identity
OK	Output keycode of last key pressed
OL	Output learn string
OM	Output memory data
ON	Output normalized (measurement —memory) data
OPDO	Output measured detector amplitude offset
OPxx	Output interrogated parameter value xx= AF, BW, DA, DB, DC, DR, RL, RP, SD, SL, SO, SP, SR, SS, ST
OR	Output rotary knob value (—32768 ≤ value ≤ +32767)
OS	Output status bytes
OT1m	Control output #1 on/off
OT2m	Control output #2 on/off
OV	Output CW value
P1	Plot channel 1 trace on external plotter
P2	Plot channel 2 trace on external plotter
P3	Plot channel 3 trace on external plotter[1]
P4	Plot channel 4 trace on external plotter[1]
PA	Plot all on external plotter
PBm	System interface control on/off
PC	Plot labels on external plotter
PD	Plot custom plot
PG	Plot grid on external plotter
PR1	Print all to monochrome printer, except softkeys and CRT graphics
PR2	Print tabular display data in monochrome
PR3	Print tabular marker/cursor data to external printer
PR4	Print all to color printer, except softkeys and CRT graphics[1]
PTd	Passthrough address set to d
PWRA	Execute a detector A power calibration
PWRB	Execute a detector B power calibration
PWRC	Execute a detector C power calibration[2]
PWRR	Execute a detector R power calibration
R1	R/A ratio measurement
R2	R/B ratio measurement

1. HP 8757D only
2. HP 8757D Option 001 only
3. Revision 3.1 or above for HP 8757E.
4. HP 8340, HP 8341, or HP 8360 series synthesized sweeper only with 8757 SYSTEM INTERFACE connected and active.
5. Limit line functions valid only for channels 1 or 2. HP 8757D only.

Table 2. Alphabetical Listing of HP 8757D/E Programming Codes (3 of 3)

Code	Action
R3	R/C ratio measurement[2]
RCn	Recall register n
RLd	Reference level set to d
RMd	Service request mask set to d
RPq	Reference position set to vertical division q
RS	Restart averaging
SCd	Set cursor to horizontal position d
SDd	Scale per division set to d
SFA	Store all instrument information to disk in file[1]
SFC	Store CRT graphics to disk in file[1]
SFD	Store data trace to disk in file[1]
SFI	Store instrument state to disk in file[1]
SFM	Store memory trace to disk in file[1]
SFN	Store normalized trace to disk in file[1]
SKq	Select softkey q: q =1 to 8
SL	Cursor search left[1]
SM	Store measurement into memory
SN	Store normalized data (measurement – memory) into memory
SOd	Smoothing set to d % of frequency span
SPd	Number of points set to d: d=101, 201, 401, 801[1], 1601[1]
SR	Cursor search right[1]
SSd	Cursor search value set to d[1]
STd	Reference level step size set to d
SUd	Specify custom plot according to d
SVn	Save register n
SW0	Non–swept mode; non–swept operation
SW1	Swept mode; normal swept operation
SW2	Sweep hold mode; non–swept mode with HP–IB bus hold off until completion of TSd
TCm	Continuous Temperature Compensation on/off
TIFs	Title for file[1]
TSd	Take d sweeps, then hold display
UP	Step up (increment)
WKs	Write softkey label
WMs	Write to channel memory.
WTs	Write title, s is an ASCII string of up to 50 characters
XAs	External detector cal value for detector A
XBs	External detector cal value for detector B
XCs	External detector cal value for detector C[2]
XRs	External detector cal value for detector R

1. HP 8757D only
2. HP 8757D Option 001 only (detector C)

NOTES:
- n = decimal integer 1 to 9
- d = variable length numeric
- m = 0 for off/1 for on
- q = unique value
- s = ASCII or binary string

For more information, call your local HP sales office listed in your telephone directory or an HP regional office listed below for the location of your nearest sales office.

United States:
Hewlett–Packard Company
4 Choke Cherry Road
Rockville, MD 20850
(301) 670–4300

Hewlett–Packard Company
5201 Tollview Drive
Rolling Meadows, IL 60008
(312) 255–9800

Hewlett–Packard Company
5161 Lankershim Blvd.
No. Hollywood, CA 91601
(818) 505–5600

Hewlett–Packard Company
2015 South Park Place
Atlanta, GA 30339
(404) 955–1500

Canada:

Hewlett–Packard Ltd.
6877 Goreway Drive
Mississauga, Ontario L4V1M8
(416) 678–9430

Australia/New Zealand:
Hewlett–Packard Australia Ltd.
31–41 Joseph Street,
Blackburn, Victoria 3130
Melbourne, Australia
(03) 895–2895

Europe/Africa/Middle East:
Hewlett–Packard S.A.
Central Mailing Department,
P.O. Box 529
1180 AM Amstelveen,
The Netherlands
(31) 20/547 9999

Far East:
Hewlett–Packard Asia Ltd.
22/F Bond Centre
West Tower
89 Queensway
Central, Hong Kong
(5) 8487777

Japan:
Yokogawa–Hewlett–Packard Ltd.
29–21, Takaido–Higashi 3–chome
Suginami–ku, Tokyo 168
(03) 331–6111

Latin America:
Latin American Region Headquarters
Monte Pelvoux Nbr. 111
Lomas de Chapultepec
11000 Mexico, D.F. Mexico
(905) 596–79–33

HEWLETT PACKARD

July 15, 1992
HP Part Number 08757–90116
Printed in U.S.A.

Addendum to HP-IB Programming Note HP 8757D/E with HP 9000 Series 200/300 computer (BASIC)

The following table provides additional codes for Table 2 of the HP-IB Programming Note, HP 8757D/E with HP 9000 Series 200/300 computer (BASIC).

Table 6-2. Alphabetical Listing of HP 8757D/E Programming Codes

Code	Action
DT	Set the date: day, month, year
OPDT	Output current date (ASCII string)
OPTM	Output current time (ASCII string
TDM	Set time: hour, minute, second
TDM	Time stamp on/off